


OpenSCAD – Употреба променљивих


Употреба променљивих

Задавање вредности димензија објеката који се креирају може се вршити преко променљивих које се дефинишу на почетку програма. На нај начин се касније, уколико је потребно, димензије објеката могу мењати променом само вредности променљивих, без промена у самом коду.

Пример: употреба променљивих

```
$fn = 50;  
radius = 5;  
height = 20;  
thickness = 1;  
difference()  
{  
  cylinder(r1 = radius, r2 = radius, h = height);  
  translate([0,0,-0.1])  
  cylinder(r1 = radius - thickness, r2 = radius - thickness, h = height + 0.2);  
}
```






OpenSCAD – Употреба променљивих


Употреба променљивих


Пример: употреба променљивих – различите вредности

`$fn = 50;`
`radius = 10;`
`height = 20;`
`thickness = 1;`



`$fn = 50;`
`radius = 10;`
`height = 15;`
`thickness = 3;`





OpenSCAD – Промена боје креираног објекта

Боја креираног објекта


Боја креираног објекта се може променити како би се у процесу пројектовања једноставније пратиле промене на дизајну. Један од начина за промену боје је

```
color( "colorname", x ) { ... }
```

Параметар **colorname** означава назив боје, а бројна вредност **x** означава транспарентност обојеног објекта, при чему је за **x = 0** објекат потпуно транспарентан, а за **x = 1** потпуно нетраспарентан.

Списак доступних боје може се видети на *Help > OpenSCAD Homepage > Color*

Напомена: Након рендеровања сви објекти су жуте боје без обзира да ли је током њиховог креирања боја била промењена!




OpenSCAD – Промена боје креираног објекта


Боја креираног објекта


Пример: Промена боје креираног објекта

```
color("Red",0.5)  
cube(10);
```



```
color("Blue",1)  
cube(10);
```







OpenSCAD – Задатак 4

Пропелер за мотор

Пројектовати 3Д модел пропелера за мотор који се састоји од централне осовине облика ваљка (црвена боја) и произвољног броја пераја која су равномерно распоређена око осовине. Дебљину осовине (3 mm), висину пропелера (20 mm), полупречник пераја (6 mm) и дебљину пераја (1 mm) као и њихов број треба унети преко променљивих на почетку програма.





OpenSCAD – Задатак 4

Пропелер за мотор – дефинисање параметара

```
$fn=50;
```

```
cylinder_radius=3; //полупречник осовине
```

```
cylinder_height=20; //висина пропелера
```


```
wing_radius = 6; //полупречник пераја
```

```
wing_thickness = 1; //дебљина пераја
```

```
wing_height = 20; //висина пераја
```

```
number_of_wings = 4; //број пераја
```


```
wing_rotation = 360/number_of_wings; //ротација пераја у степенима
```




OpenSCAD – Задатак 4

Пропелер за мотор – креирање једног пераја

```
module wing()  
{  
  difference()  
  {  
    cylinder(r1 = wing_radius, r2 = wing_radius, h = wing_height);  
    translate([0, 0, -0.1])  
    cylinder(r1 = wing_radius-wing_thickness, r2 = wing_radius-wing_thickness, h = wing_height + 0.2);  
    translate([-wing_radius, 0, -0.1])  
    cube([2*wing_radius, wing_radius, wing_height+0.2]);  
  }  
}
```





OpenSCAD – Задатак 4

Пропелер за мотор – креирање целог пропелера

```
color("Red", 1)
cylinder(r1 = cylinder_radius, r2 = cylinder_radius, h = cylinder_height);
for (i = [0:1:number_of_wings-1])
{
  rotate([0, 0, wing_rotation*i])
  translate([-wing_radius, 0, 0])
  wing();
}
```

